物理学情報処理演習

14. 数式処理　補足
Mapleで物理を再勉強

大久保晋

自由落下を計算してみよう

次のように入力する

物理定数を使えるようにする

```
> with(ScientificConstants);
```

[*AddConstant, AddElement, AddProperty, Constant, Element, GetConstant, GetConstants, GetElement, GetElements, GetError, GetIsotopes, GetProperties, GetProperty, GetUnit, GetValue, HasConstant, HasElement, HasProperty, ModifyConstant, ModifyElement*]

ニュートンの方程式

```
> falleq:= m* diff(y(t), t$2) = - g * m ;
```

２回 t で微分する

$$falleq := m\left(\frac{d^2}{dt^2}y(t)\right) = -g\,m$$

初期値 高さ1000から
初速度０から落とす

```
> fallsol:= dsolve([falleq, y(0)=1000, D(y)(0)=0], y(t));
```

$$fallsol := y(t) = -\frac{1}{2}g\,t^2 + 1000$$

初期値を代入し計算させ、fallsolとした

```
> g:= Constant('g');
```

$$g := Constant(g)$$

重力加速度を定数表から出す

```
> falleq2:= eval( y(t), subs( {g= evalf(g)},fallsol));
```

$$falleq2 := -4.903325000\,t^2 + 1000$$

```
> plot(falleq2, t=0..10, title="free fall", labels=["time", "y(t)"],
  font=[times,ROMAN,10]);
```

図を書いてみた。
きちんと落ちていくだろうか？

電磁気学２　LC回路

右にあるようなLとCで構成される
閉回路の電荷の変化を計算してみよう



閉回路の法則より

$$\phi(t) - L\frac{dI(t)}{dt} - \frac{Q(t)}{C} = 0 \qquad I(t) = \frac{dQ(t)}{dt}$$

なので、求める微分方程式は

$$L\frac{d^2Q(t)}{dt^2} + \frac{Q(t)}{C} = \phi(t)$$

ここで

$$\phi(t) = \phi_0 \cos\omega t$$

解は

$$I(t) = \alpha\cos\omega t + \beta sin\omega t = \mathrm{Re}(I_0 e^{i\beta} \cdot e^{i\omega t})$$

Mapleで次のように入力してみよう

```
> with(DEtools);
```
←微分積分のツールを使う宣言

```
> LCcircuit:= L* diff(Q(t), t$2) + (1/C)* Q(t) = p * cos(w*t);
```
←問題の２階微分方程式

$$LCcircuit := L\left(\frac{d^2}{dt^2}Q(t)\right) + \frac{Q(t)}{C} = p\cos(w\,t)$$

２回 t で微分する

```
> dsolve(LCcircuit, Q(t));
```

$$Q(t) = \sin\left(\frac{t}{\sqrt{C}\sqrt{L}}\right)\_C2 + \cos\left(\frac{t}{\sqrt{C}\sqrt{L}}\right)\_C1 - \frac{p\cos(w\,t)\,C}{-1 + w^2 C L}$$

微分方程式を解け

```
> LCcircuit2:=subs( {p=0, L=1, C=1, w=2*Pi}, LCcircuit);
```

$$LCcircuit2 := \left(\frac{d^2}{dt^2}Q(t)\right) + Q(t) = 0$$

パラメーターを代入し、LCcircuit2とした
p=0 として、外部から交流電流は流さない
L=1, C=1, ω=2π

```
> dsolve( {LCcircuit2, Q(0)=1, D(Q)(0)=0}, Q(t));
```

$$Q(t) = \cos(t)$$

境界条件、Q(0)=1, dQ(0)/dt =0 としてQ(t)を解いた

```
> solLCcircuit2:= eval(Q(t), dsolve( {LCcircuit2, Q(0)=1, D(Q)(0)=0}, Q(t)));
```

$$solLCcircuit2 := \cos(t)$$

境界条件までいれて、Q(t)をといて、それを式solLCcircuit2とした

```
> plot(solLCcircuit2, t=0..4*Pi);
```

solLCcircuit2を t=0から4πまで作画した



適当にパラメーターをいじると変化することがわかる

```
LCcircuit3:=subs( {p=1, L=1, C=1, w=0.5*Pi}, LCcircuit);
solLCcircuit3:= eval(Q(t), dsolve( {LCcircuit3, Q(0)=1, D(Q)(0)=0}, Q(t)));
plot( solLCcircuit3, t=0..4*Pi);
```

このようにすれば、パラメーターを変化させることができる。
この値をいれてみて、グラフを書きもとのcos波と変わったことを確認しよう。